Riemann Curvature:

The Riemann tensor is given by

$$R^{\rho}_{\sigma\mu\nu} := \partial_\mu \Gamma^{\rho}_{\nu\sigma} - \partial_\nu \Gamma^{\rho}_{\mu\sigma} + \Gamma^{\rho}_{\mu\lambda}\Gamma^{\lambda}_{\nu\sigma} - \Gamma^{\rho}_{\nu\lambda}\Gamma^{\lambda}_{\mu\sigma}$$

where the $\Gamma^{i}{}_{jk}$ is the Christoffel symbol of second kind, which is given by

$$\Gamma^{\alpha}_{\mu\nu} = \frac{1}{2} g^{\alpha\delta}\left(-\partial_\delta g_{\mu\nu} + \partial_\mu g_{\delta\nu} + \partial_\nu g_{\delta\nu}\right)$$

On the other hand, the Ricci tensor is defined as the contraction with the Riemann tensor, this is

$$R_{\mu\nu} = R^{\rho}_{\mu\rho\nu}$$

and the Ricci scalar is defined in terms of the Ricci tensor as follows

$$R = g^{\mu\nu} R_{\mu\nu}$$

The first step is to define the metric tensor its inverse, and the set of coordinates that we will work on.

In[1]:=
```
g = {{1 + x^2/(L^2 - x^2 - y^2), (x y/(L^2 - x^2 - y^2))}, {(x y/(L^2 - x^2 - y^2)), 1 + y^2/(L^2 - x^2 - y^2)}};
ginv = Inverse[g];
dim = Dimensions[g][[1]];
coords = {x, y};
```

Let’s display the metric and it’s inverse:

In[5]:=
```
g // MatrixForm
```

Out[5]//MatrixForm=

$$\begin{pmatrix} 1 + \frac{x^2}{L^2-x^2-y^2} & \frac{x\,y}{L^2-x^2-y^2} \\ \frac{x\,y}{L^2-x^2-y^2} & 1 + \frac{y^2}{L^2-x^2-y^2} \end{pmatrix}$$

In[6]:=
```
Simplify[ginv] // MatrixForm
```

Out[6]//MatrixForm=

$$\begin{pmatrix} 1 - \frac{x^2}{L^2} & -\frac{x\,y}{L^2} \\ -\frac{x\,y}{L^2} & 1 - \frac{y^2}{L^2} \end{pmatrix}$$

Now, let’s compute the Christoffel symbol of the second kind:

In[7]:=
```
Γ = Table[Simplify[1/2 Sum[
      ginv⟦α, δ⟧ × (
        -D[g⟦μ, ν⟧, coords⟦δ⟧] + D[g⟦δ, ν⟧, coords⟦μ⟧] + D[g⟦δ, μ⟧, coords⟦ν⟧]),
      {δ, 1, dim}]], {α, 1, dim}, {μ, 1, dim}, {ν, 1, dim}]
```

Out[7]=

$$\left\{\left\{\left\{\frac{x\left(L^2-y^2\right)}{L^2\left(L^2-x^2-y^2\right)},\frac{x^2\,y}{L^2\left(L^2-x^2-y^2\right)}\right\},\left\{\frac{x^2\,y}{L^2\left(L^2-x^2-y^2\right)},\frac{x\left(L^2-x^2\right)}{L^2\left(L^2-x^2-y^2\right)}\right\}\right\},\left\{\left\{\frac{y\left(L^2-y^2\right)}{L^2\left(L^2-x^2-y^2\right)},\frac{x\,y^2}{L^2\left(L^2-x^2-y^2\right)}\right\},\left\{\frac{x\,y^2}{L^2\left(L^2-x^2-y^2\right)},\frac{\left(L^2-x^2\right)y}{L^2\left(L^2-x^2-y^2\right)}\right\}\right\}\right\}$$

Since this object has 3 indices, and the number of dimensions is 2, the shape of this object is expected to be {3,3,3}, let’s check that:

In[8]:=
```
Dimensions[Γ] == {4, 4, 4}
```

Out[8]=
```
False
```

Great, now let’s move on let’s compute the Riemann tensor:

In[9]:=
```
rTensor = Table[
  Simplify[
   D[Γ⟦ρ, ν, σ⟧, coords⟦μ⟧] - D[Γ⟦ρ, μ, σ⟧, coords⟦ν⟧] +
    Sum[Γ⟦ρ, μ, λ⟧ × Γ⟦λ, ν, σ⟧, {λ, 1, dim}] - Sum[Γ⟦ρ, ν, λ⟧ × Γ⟦λ, μ, σ⟧, {λ, 1, dim}]
  ], {ρ, 1, dim}, {σ, 1, dim}, {μ, 1, dim}, {ν, 1, dim}]
```

Out[9]=

$$\left\{\left\{\left\{\left\{0,\frac{x\,y}{L^2\left(L^2-x^2-y^2\right)}\right\},\left\{-\frac{x\,y}{L^2\left(L^2-x^2-y^2\right)},0\right\}\right\},\left\{\left\{0,\frac{L^2-x^2}{L^2\left(L^2-x^2-y^2\right)}\right\},\left\{\frac{-L^2+x^2}{L^2\left(L^2-x^2-y^2\right)},0\right\}\right\}\right\},\left\{\left\{\left\{0,\frac{-L^2+y^2}{L^2\left(L^2-x^2-y^2\right)}\right\},\left\{\frac{L^2-y^2}{L^2\left(L^2-x^2-y^2\right)},0\right\}\right\},\left\{\left\{0,-\frac{x\,y}{L^2\left(L^2-x^2-y^2\right)}\right\},\left\{\frac{x\,y}{L^2\left(L^2-x^2-y^2\right)},0\right\}\right\}\right\}\right\}$$

Since this tensor does not vanish for all its components, we can conclude that the space is not flat. Even more, this object is a good tensor, and the shape of this object is expected to be {2,2,2,2}, so let’s check again for this

In[10]:=
```
Dimensions[rTensor] == {4, 4, 4, 4}
```

Out[10]=
```
False
```

Now let’s compute the version of the Riemann tensor with all the indices down:

In[11]:=
```
rDown = Table[Simplify[Sum[g⟦i, m⟧ × rTensor⟦m, j, k, l⟧, {m, 1, dim}]],
   {i, 1, dim}, {j, 1, dim}, {k, 1, dim}, {l, 1, dim}]
```

Out[11]=

$$\left\{\left\{\{\{0,0\},\{0,0\}\},\left\{\left\{0,\frac{1}{L^2-x^2-y^2}\right\},\left\{\frac{1}{-L^2+x^2+y^2},0\right\}\right\}\right\},\left\{\left\{\left\{0,\frac{1}{-L^2+x^2+y^2}\right\},\left\{\frac{1}{L^2-x^2-y^2},0\right\}\right\},\{\{0,0\},\{0,0\}\}\right\}\right\}$$

And as we can see, the shape remains the same, as expected:

In[12]:=
```
Dimensions[rDown]
```

Out[12]=

{2, 2, 2, 2}

And we can also compute the Ricci tensor:

In[13]:=
```
ricci = Table[Sum[rTensor⟦ρ, μ, ρ, ν⟧, {ρ, 1, dim}], {μ, 1, dim}, {ν, 1, dim}]
```

Out[13]=

$$\left\{\left\{\frac{L^2-y^2}{L^2\left(L^2-x^2-y^2\right)},\frac{x\,y}{L^2\left(L^2-x^2-y^2\right)}\right\},\left\{\frac{x\,y}{L^2\left(L^2-x^2-y^2\right)},\frac{L^2-x^2}{L^2\left(L^2-x^2-y^2\right)}\right\}\right\}$$

And check that indeed it has the right shape:

In[14]:=
```
Dimensions[ricci] == {4, 4}
```

Out[14]=

False

Finally, we can compute the Ricci scalar:

In[15]:=
```
ricciScalar = Simplify[Sum[ginv⟦μ, ν⟧ × ricci⟦μ, ν⟧, {μ, 1, dim}, {ν, 1, dim}]]
```

Out[15]=

$$\frac{2}{L^2}$$